

NEWPAGE

<pre> NEWPAGE [(rep)] [[IF] EVEN [IF] TOP [OF] [PAGE] [WHEN] LESS [THAN] operand1 [LINES] [LEFT] [[WITH] TITLE] (see WRITE TITLE syntax)] </pre>
--

Operand	Possible Structure		Possible Formats										Referencing Permitted	Dynamic Definition			
Operand1	C	S														yes	no

Related Statements: AT TOP OF PAGE | AT END OF PAGE

Function

The NEWPAGE statement is used to cause an advance to a new page. NEWPAGE also causes any AT END OF PAGE and WRITE TRAILER statements to be executed. A default title containing the date, time of day, and page number will appear on each new page unless a WRITE TITLE, WRITE NOTITLE, or DISPLAY NOTITLE statement is specified to define specific title processing.

Notes:

The advance to a new page is not performed at the time when the NEWPAGE statement is executed. It is performed only when a subsequent statement which produces output is executed.

If the NEWPAGE statement is not used, page advance is controlled automatically based on the session parameter PS.

Report Specification - rep

The notation (*rep*) may be used to specify the identification of the report for which the NEWPAGE statement is applicable. A value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified.

If (*rep*) is not specified, the NEWPAGE statement will be applicable to the first report (report 0).

EVEN IF TOP OF PAGE

This option is used to cause a new page (with corresponding AT TOP OF PAGE and page title processing) to be generated, even if a new page was initiated immediately before the NEWPAGE statement was encountered.

WHEN LESS THAN operand1 LINES LEFT

This option is used to cause a new page to be generated when there are less than *operand1* lines left on the current page (current line count compared with value for session parameter PS).

WITH TITLE

The WITH TITLE option may be used to specify a title which is to be written to the new page generated. The title is specified using the same syntax as described for the WRITE TITLE statement, except that the SKIP clause in a NEWPAGE WITH TITLE statement is not allowed.

Example

```

/* EXAMPLE 'NWPEX1S': NEWPAGE
/*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 NAME
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
/*****
LIMIT 15
READ EMPLOY-VIEW BY CITY FROM 'DENVER'
  DISPLAY CITY (IS=ON) NAME SALARY (1) CURR-CODE (1)
  AT BREAK OF CITY
  SKIP 1
/*****
  NEWPAGE WHEN LESS THAN 10 LINES LEFT
  WRITE '*****'
    / 'SUMMARY FOR ' OLD(CITY)
    / '*****'
    / '*****'
    / 'SUM OF SALARIES:' SUM(SALARY(1))
    / 'AVG OF SALARIES:' AVER(SALARY(1))
    / '*****'
  NEWPAGE
/*****
  END-BREAK
END-READ
END

```

PAGE	1			03-03-27	15:21:13
	CITY	NAME	ANNUAL SALARY	CURRENCY CODE	
	-----	-----	-----	-----	-----
	DENVER	TANIMOTO	33000	USD	
		MEYER	50000	USD	

	SUMMARY FOR DENVER				

	SUM OF SALARIES:	83000			
	AVG OF SALARIES:	41500			

PAGE	2			03-03-27	15:21:13
	CITY	NAME	ANNUAL SALARY	CURRENCY CODE	
	-----	-----	-----	-----	-----
	DERBY	DEAKIN	7950	UKL	
		GARFIELD	7200	UKL	
		MUNN	8200	UKL	
		MUNN	5200	UKL	
		GREBBY	8450	UKL	
		WHITT	7450	UKL	
		PONSONBY	4450	UKL	
		MAGUIRE	4000	UKL	
		HEYWOOD	3800	UKL	
		BRYDEN	6450	UKL	
		SMITH	29000	UKL	
		CONQUEST	34000	UKL	
		ACHIESON	10500	UKL	

PAGE	3			03-03-27	15:21:13
	CITY	NAME	ANNUAL SALARY	CURRENCY CODE	

SUMMARY FOR DERBY					

SUM OF SALARIES: 136650					
AVG OF SALARIES: 10511					

Equivalent reporting-mode example: See program NWPEX1R in library SYSEXRM.